
Change 1 
	Function name
	fl_TC_9_2_2_1_9_Body

	Reason for change
	In current implementation, when EUTRAN Handover is performed:

Some UEs send TAU request at the TTI as RRCConnectionReconfigurationComplete message for handover completion. The problem in current implementation is that the TAU request in this case is unexpected message in TTCN because TTCN performs functions inside f_EUTRA_SS_SRBs_DRBs_Reset and some of them require confirmations (f_EUTRA_SS_PdcpCount_GetAll).

Suggested solution is to skip step 13-14 of 36.523-3 cl. 7.16.1 in case where TAU procedure is expected in TC and execute these steps after the TAU procedure is completed.

	Summary of change
	Added input parameter to f_EUTRA_508RRC_IntraLTE_HO_InterCell to wait for TAU procedure
Added function f_EUTRA_508RRC_IntraLTE_HO_InterCell_Common_step13to14  after TAU procedure is finished

	TTCN module
	\LTE\9_2\NAS_Detach.ttcn

	MCC160 Comment
	


Before change

    function fl_TC_9_2_2_1_9_Body() runs on EUTRA_PTC

  { // @sic R5s100883 sic@

    var NAS_MSG_Indication_Type v_NasInd;

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

    var template (value) CellPowerList_Type v_CellPowerList;

    var GutiParameters_Type v_GutiParams_B := f_EUTRA_CellInfo_GetGuti (eutra_CellB);

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams_B.PLMN_Identity);

    var template (omit) EPS_BearerContextStatus v_EpsBearerCtxtStatus;

    var float v_NoReactionTime := f_EUTRA_SetTimerToleranceMax( eutra_CellB, rrcTimer, 3.0 );

    var TrackingAreaCode v_Tac := f_EUTRA_CellInfo_GetTAC ( eutra_CellB );

    var NAS_Lac v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode ( eutra_CellB );

    var template (omit) LocAreaId v_LAI := f_GetLAI ( v_PLMN, v_LAC, NORMAL); // @sic R5s110176 sic@

    var template (omit) MobileIdentity v_MSId := f_GetMSId (NORMAL); // @sic R5s110176 sic@

    var template (value) TrackingAreaIdList v_TaiList := cds_TAIListNonConsecutive_tlv ( v_PLMN, { bit2oct(v_Tac) } );

    var EPS_UpdateTypeValue v_UpdateType := f_GetEPSTAUType(NORMAL); // @sic R5s110176 sic@

    var SRB_COMMON_IND v_ReceivedAsp;

    //  Cause UE to initiate detach from the EPS services.

    //@siclog "Step 1" siclog@

    f_EUTRA_UE_Detach_SwitchOffUe_Trigger(eutra_CellA, detach); // @sic R5s150320 sic@

    //+   UE may optionally perform IMS de-registration using the generic procedure defined in 34.229-1 [35] Annex C.30.

    //@siclog "Step 1A" siclog@

    //       The UE transmits a DETACH REQUEST message with the Detach type IE indicating

    //       'EPS detach' and 'normal detach'? (The UE starts timer T3421).

    //@siclog "Step 2" siclog@

    f_EUTRA_UE_DetachMO(eutra_CellA, RRC_CONNECTED, -, cr_DetachType('0'B, (tsc_DetachMO_EpsOnly, tsc_DetachMO_Combined))); // @sic R5s150190 r2 sic@

    // The SS does not respond to the DETACH REQUEST message.

    // Set the cell type of cell A to the "Non-Suitable cell".

    //         Set the cell type of cell B to the "Serving cell".

    // Note: T3421 value is specified as 15s in TS 24.301 and it is assumed that

    //       SS can configure cells within this time.

    //@siclog "Step 3 - 4" siclog@

    v_CellPowerList := {

      cs_CellPower ( eutra_CellA, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE ),

      cs_CellPower ( eutra_CellB, tsc_ServingCellRS_EPRE )

    };

    f_EUTRA_SetCellPowerList ( v_CellPowerList );

    //SS now creates a handover situation to direct the UE to cellB

    //@siclog "Step 4A - 4B" siclog@

    f_EUTRA_508RRC_IntraLTE_HO_InterCell ( eutra_CellA, eutra_CellB );

    //  Check: Does the UE transmit a TRACKING AREA UPDATE REQUEST message?

    //@siclog "Step 5" siclog@

    SRB.receive( car_SRB2_NasPdu_IND(eutra_CellB,

                                     cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                       cdr_TAU_RequestAnyActiveFlag(v_UpdateType,

                                                                                    f_EUTRA_SecurityKSIasme_Get(),

                                                                                    cr_DRXparameter_Any ('5C'O) ifpresent, // @sic R5s120210, R5-123117 sic@

                                                                                    v_AdditionalUpdateType)))) -> value v_ReceivedAsp;

    v_NasInd := v_ReceivedAsp.Signalling.Nas[0] ;

    v_EpsBearerCtxtStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus;

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5: TAU Request sent by UE");

    // The SS transmits a TRACKING AREA UPDATE ACCEPT message.

    //@siclog "Step 6" siclog@

    SRB.send ( cas_SRB2_NasPdu_REQ ( eutra_CellB,

                                     cs_TimingInfo_Now,

                                     cs_NAS_Request ( tsc_SHT_IntegrityProtected_Ciphered,

                                                      cs_508_TAU_Accept ( v_UpdateType,

                                                                          f_GutiParameters2MobileIdentity (tsc_IEI_Guti, v_GutiParams_B ), //GUTI-2

                                                                          v_TaiList,

                                                                          v_EpsBearerCtxtStatus,

                                                                          v_LAI,

                                                                          v_MSId,

                                                                          f_GetAdditionalUpdateResult(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType)))));

    // Check: Does the UE transmit a TRACKING AREA UPDATE COMPLETE message?

    //@siclog "Step 7" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellB,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_TAU_Complete) ));

    //  The UE transmits a DETACH REQUEST message with the Detach type IE indicating

    //         'EPS detach' and 'normal detach'? The UE starts timer T3421.

    //@siclog "Step 8" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellB,

                                    cr_NAS_Indication (tsc_SHT_IntegrityProtected_Ciphered,

                                                       cr_DETACH_REQUEST_NormalDetachandMO (cr_NAS_KeySetIdentifier (?, tsc_NasKsi_NativeSecurityContext),

                                                                                            f_GutiParameters2MobileIdentity(omit, v_GutiParams_B)))));  // @sic R5-113807 sic@

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8: Detach Request sent by UE");

    // SS responds with DETACH ACCEPT message

    //@siclog "Step 9" siclog@

    SRB.send (cas_SRB2_NasPdu_REQ ( eutra_CellB,

                                    cs_TimingInfo_Now,

                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                   cs_508_DETACH_ACCEPT) ));

    f_EUTRA_RRC_ConnectionRelease(eutra_CellB);

    //  Check: Does the test result of the generic procedure in TS 36.508 subclause 6.4.2.5 indicates that

    // the UE responds to paging when paged with GUTI-2 and with CN domain indicator set to "PS"?

    //@siclog "Step 10" siclog@

    f_EUTRA_508Check_NASNoResponse ( eutra_CellB,

                                     cs_NasPaging_OneRecord ( cs_PagingUE_Identity_S_TMSI ( v_GutiParams_B.MME_Code, v_GutiParams_B.M_TMSI )),

                                     v_NoReactionTime );

  }// end function fl_TC_9_2_2_1_9_Body
 After change

  function fl_TC_9_2_2_1_9_Body() runs on EUTRA_PTC

  { // @sic R5s100883 sic@

    var NAS_MSG_Indication_Type v_NasInd;

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

    var template (value) CellPowerList_Type v_CellPowerList;

    var GutiParameters_Type v_GutiParams_B := f_EUTRA_CellInfo_GetGuti (eutra_CellB);

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams_B.PLMN_Identity);

    var template (omit) EPS_BearerContextStatus v_EpsBearerCtxtStatus;

    var float v_NoReactionTime := f_EUTRA_SetTimerToleranceMax( eutra_CellB, rrcTimer, 3.0 );

    var TrackingAreaCode v_Tac := f_EUTRA_CellInfo_GetTAC ( eutra_CellB );

    var NAS_Lac v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode ( eutra_CellB );

    var template (omit) LocAreaId v_LAI := f_GetLAI ( v_PLMN, v_LAC, NORMAL); // @sic R5s110176 sic@

    var template (omit) MobileIdentity v_MSId := f_GetMSId (NORMAL); // @sic R5s110176 sic@

    var template (value) TrackingAreaIdList v_TaiList := cds_TAIListNonConsecutive_tlv ( v_PLMN, { bit2oct(v_Tac) } );

    var EPS_UpdateTypeValue v_UpdateType := f_GetEPSTAUType(NORMAL); // @sic R5s110176 sic@

    var SRB_COMMON_IND v_ReceivedAsp;

    //  Cause UE to initiate detach from the EPS services.

    //@siclog "Step 1" siclog@

    f_EUTRA_UE_Detach_SwitchOffUe_Trigger(eutra_CellA, detach); // @sic R5s150320 sic@

    //+   UE may optionally perform IMS de-registration using the generic procedure defined in 34.229-1 [35] Annex C.30.

    //@siclog "Step 1A" siclog@

    //       The UE transmits a DETACH REQUEST message with the Detach type IE indicating

    //       'EPS detach' and 'normal detach'? (The UE starts timer T3421).

    //@siclog "Step 2" siclog@

    f_EUTRA_UE_DetachMO(eutra_CellA, RRC_CONNECTED, -, cr_DetachType('0'B, (tsc_DetachMO_EpsOnly, tsc_DetachMO_Combined))); // @sic R5s150190 r2 sic@

    // The SS does not respond to the DETACH REQUEST message.

    // Set the cell type of cell A to the "Non-Suitable cell".

    //         Set the cell type of cell B to the "Serving cell".

    // Note: T3421 value is specified as 15s in TS 24.301 and it is assumed that

    //       SS can configure cells within this time.

    //@siclog "Step 3 - 4" siclog@

    v_CellPowerList := {

      cs_CellPower ( eutra_CellA, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE ),

      cs_CellPower ( eutra_CellB, tsc_ServingCellRS_EPRE )

    };

    f_EUTRA_SetCellPowerList ( v_CellPowerList );

    //SS now creates a handover situation to direct the UE to cellB

    //@siclog "Step 4A - 4B" siclog@

    f_EUTRA_508RRC_IntraLTE_HO_InterCell ( eutra_CellA, eutra_CellB, -, -, -, -, -, -, -, -, -, true ); 

    //  Check: Does the UE transmit a TRACKING AREA UPDATE REQUEST message?

    //@siclog "Step 5" siclog@

    SRB.receive( car_SRB2_NasPdu_IND(eutra_CellB,

                                     cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                       cdr_TAU_RequestAnyActiveFlag(v_UpdateType,

                                                                                    f_EUTRA_SecurityKSIasme_Get(),

                                                                                    cr_DRXparameter_Any ('5C'O) ifpresent, // @sic R5s120210, R5-123117 sic@

                                                                                    v_AdditionalUpdateType)))) -> value v_ReceivedAsp;

    v_NasInd := v_ReceivedAsp.Signalling.Nas[0] ;

    v_EpsBearerCtxtStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus;

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5: TAU Request sent by UE");

    f_EUTRA_508RRC_IntraLTE_HO_InterCell_Common_step13to14(eutra_CellA, eutra_CellB); 

    // The SS transmits a TRACKING AREA UPDATE ACCEPT message.

    //@siclog "Step 6" siclog@

    SRB.send ( cas_SRB2_NasPdu_REQ ( eutra_CellB,

                                     cs_TimingInfo_Now,

                                     cs_NAS_Request ( tsc_SHT_IntegrityProtected_Ciphered,

                                                      cs_508_TAU_Accept ( v_UpdateType,

                                                                          f_GutiParameters2MobileIdentity (tsc_IEI_Guti, v_GutiParams_B ), //GUTI-2

                                                                          v_TaiList,

                                                                          v_EpsBearerCtxtStatus,

                                                                          v_LAI,

                                                                          v_MSId,

                                                                          f_GetAdditionalUpdateResult(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType)))));

    // Check: Does the UE transmit a TRACKING AREA UPDATE COMPLETE message?

    //@siclog "Step 7" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellB,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_TAU_Complete) ));

    //  The UE transmits a DETACH REQUEST message with the Detach type IE indicating

    //         'EPS detach' and 'normal detach'? The UE starts timer T3421.

    //@siclog "Step 8" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellB,

                                    cr_NAS_Indication (tsc_SHT_IntegrityProtected_Ciphered,

                                                       cr_DETACH_REQUEST_NormalDetachandMO (cr_NAS_KeySetIdentifier (?, tsc_NasKsi_NativeSecurityContext),

                                                                                            f_GutiParameters2MobileIdentity(omit, v_GutiParams_B)))));  // @sic R5-113807 sic@

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8: Detach Request sent by UE");

    // SS responds with DETACH ACCEPT message

    //@siclog "Step 9" siclog@

    SRB.send (cas_SRB2_NasPdu_REQ ( eutra_CellB,

                                    cs_TimingInfo_Now,

                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                   cs_508_DETACH_ACCEPT) ));

    f_EUTRA_RRC_ConnectionRelease(eutra_CellB);

    //  Check: Does the test result of the generic procedure in TS 36.508 subclause 6.4.2.5 indicates that

    // the UE responds to paging when paged with GUTI-2 and with CN domain indicator set to "PS"?

    //@siclog "Step 10" siclog@

    f_EUTRA_508Check_NASNoResponse ( eutra_CellB,

                                     cs_NasPaging_OneRecord ( cs_PagingUE_Identity_S_TMSI ( v_GutiParams_B.MME_Code, v_GutiParams_B.M_TMSI )),

                                     v_NoReactionTime );

  }// end function fl_TC_9_2_2_1_9_Body
Change 2 
	Function name
	f_EUTRA_508RRC_IntraLTE_HO_InterCell_Common

	Reason for change
	In current implementation, when EUTRAN Handover is performed:

Some UEs send TAU request at the TTI as RRCConnectionReconfigurationComplete message for handover completion. The problem in current implementation is that the TAU request in this case is unexpected message in TTCN because TTCN performs functions inside f_EUTRA_SS_SRBs_DRBs_Reset and some of them require confirmations (f_EUTRA_SS_PdcpCount_GetAll).

Suggested solution is to skip step 13-14 of 36.523-3 cl. 7.16.1 in case where TAU procedure is expected in TC and execute these steps after the TAU procedure is completed.

	Summary of change
	Steps 13 and 14 of f_EUTRA_508RRC_IntraLTE_HO_InterCell_Common moved to new function f_EUTRA_508RRC_IntraLTE_HO_InterCell_Common_step13to14

	TTCN module
	Common\EUTRA\EUTRA_RRCSteps.ttcn

	MCC160 Comment
	


Before change

    
function f_EUTRA_508RRC_IntraLTE_HO_InterCell_Common(EUTRA_CellId_Type                         p_SourceCellId,

                                                       EUTRA_CellId_Type                         p_TargetCellId,

                                                       template (value) DL_DCCH_Message          p_RRC_ConnRecongMsg,

                                                       template (value) RachProcedureConfig_Type p_RachProcedureConfig,

                                                       template (value) RadioBearerList_Type     p_DrbConfigList := cs_DrbConfigList_DRB1,

                                                       C_RNTI                                    p_C_RNTI        := tsc_C_RNTI_Def2,

                                                       template (omit)  DciUlInfo_Type           p_DciUlInfo_TargetCell      := cs_DciInfo_CcchDcchDtchUL_Def,

                                                       template (value) UL_GrantConfig_Type      p_UL_GrantConfig_SourceCell := cs_UL_GrantConfig_Def,

                                                       template (value) UL_GrantConfig_Type      p_UL_GrantConfig_TargetCell := cs_UL_GrantConfig_Def,

                                                       template (present) UL_DCCH_Message        p_RRCConnReconfigComplete := cr_508_RRCConnectionReconfigurationComplete)

    runs on EUTRA_PTC

  { /* @sic R5s120709 function introduced to cope with "Full Handover" as needed for TC_13_4_1_5 sic@ */

    /* @sic R5s140147: additional parameter p_RRCConnReconfigComplete sic@ */

    var DRB_IdentityList_AllAndAMonly_Type v_DRB_IdentityList;

    var EUTRA_SecurityParams_Type v_Auth_Params := f_EUTRA_Security_Get();

    var NextHopChainingCount v_NextHopChainingCount := 0; //According to 36.508 cl. 4.6.1 RRCConnectionReestablishment

    var template (omit) MeasConfig v_MeasConfig := p_RRC_ConnRecongMsg.message_.c1.rrcConnectionReconfiguration.criticalExtensions.c1.rrcConnectionReconfiguration_r8.measConfig;

    var boolean v_MeasGapConfigIsPresent := isvalue(v_MeasConfig) and ispresent(v_MeasConfig.measGapConfig);

    var DrxCtrl_Type v_DrxCtrl;

    var boolean v_CnfFlag := tsc_NoCnfReq;  /* @sic R5s150922 - Additional changes: v_CnfFlag instead of v_CnfFlag sic@ */

    // Build DRB identity list from inputed radiobearer config list

    v_DRB_IdentityList := f_EUTRA_DRB_IdentityList(p_DrbConfigList);

    // Step 0: Configuration of DRBs at the target cell

    f_EUTRA_SS_CommonRadioBearerConfig(p_TargetCellId, p_DrbConfigList);       /* @sic R5s150338 - dynamic DRB configuration sic@ */

    // Step 1: Transfer of the PDCP Count for AM DRBs from source to target cell   @sic R5-120721 change 7 sic@

    f_EUTRA_SS_PdcpCount_Handover(p_SourceCellId, p_TargetCellId, v_DRB_IdentityList.AMonly);

    // Step 2: Tell the SS about the HO    @sic R5w100212 sic@

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetCellId, cas_PdcpHandoverInit_REQ(p_SourceCellId, p_TargetCellId)); // @sic R5s100556, R5s100560 sic@

    // Step 3: Configure C-RNTI based contention resolution in target cell

    f_EUTRA_SS_ConfigRachProcedure(p_TargetCellId, p_C_RNTI, p_RachProcedureConfig);

    // Save new C-RNTI in cell configuration for p_TargetCellId

    f_EUTRA_CellInfo_SetC_RNTI(p_TargetCellId, p_C_RNTI);

    // Step 4: activate RRC security at SS target cell for SRB 1, 2 and 1 AM DRB

    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest(p_TargetCellId,

                                                             v_Auth_Params,

                                                             v_NextHopChainingCount,

                                                             -,

                                                             v_DRB_IdentityList.AMonly);

    f_EUTRA_Security_Set(v_Auth_Params);

    // Step 5: configure DRX and MeasGap at the target cell (if necessary)

    if (f_EUTRA_CellInfo_DrxIsConfigured(p_SourceCellId)) {

      v_DrxCtrl := f_EUTRA_CellInfo_GetDrxCtrl(p_SourceCellId);

      f_EUTRA_SS_DrxCtrlConfig(p_TargetCellId, v_DrxCtrl);

    }

    if (v_MeasGapConfigIsPresent) {

      f_EUTRA_SS_MeasGapCtrlConfig(p_TargetCellId, cs_MeasGapCtrl_Config(valueof(v_MeasConfig.measGapConfig)));   // valueof to avoid compiler warnings: ok as v_MeasConfig.measGapConfig is checked to be present

    }

    // Step 6: stop periodic TA on source cell   @sic R5s100520 change 10 sic@

    f_EUTRA_SS_CommonCellConfig(p_SourceCellId, cas_ULGrantAllocation_REQ(p_SourceCellId, cs_TimingInfo_Now, -, cs_PUCCH_Synch_None, p_UL_GrantConfig_SourceCell));

    // Step 7: configure UL grant configuration on target cell    @sic R5s100515 sic@

    f_EUTRA_SS_CommonCellConfig(p_TargetCellId, cas_ULGrantAllocation_REQ(p_TargetCellId, cs_TimingInfo_Now, p_DciUlInfo_TargetCell, cs_PUCCH_Synch_None, p_UL_GrantConfig_TargetCell));

    f_IP_ChangeEutraCell(IP, p_TargetCellId);      /* @sic R5s150031 Change 1 sic@ */

    // Step 8: Transmit RRCConnectionReconfiguration/HO command to UE

    SRB.send(cas_SRB1_RrcPdu_REQ(p_SourceCellId, cs_TimingInfo_Now, p_RRC_ConnRecongMsg));

    // Step 9:

    SRB.receive(car_SRB1_RrcPdu_IND(p_TargetCellId, p_RRCConnReconfigComplete));       // @sic R5s100430  Change 2, R5s140147 sic@

    // Step 10: configure periodic TA on target cell    @sic R5s100525 change 1.3 sic@

    f_EUTRA_SS_CommonCellConfig(p_TargetCellId, cas_PUCCH_Synch_Config_REQ(p_TargetCellId, cs_TimingInfo_Now, cs_PUCCH_Synch_Auto_Def, tsc_NoCnfReq)); // @sic R5s100801 sic@

    // Step 11: Tell the SS about the HO    @sic R5w100212 sic@

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetCellId, cas_PdcpHandoverComplete_REQ(p_TargetCellId)); // @sic R5s100556, R5s100560 sic@

    // Step 12: configure RACH procedure for target cell back to default    @sic R5s100520 change 11 sic@

    f_EUTRA_SS_ConfigRachProcedure_Def(p_TargetCellId, -, -, -, -, -, v_CnfFlag);     /* @sic acc. to R5s100029 change 3 sic@ */

    // Step 13: Re-establish SRBs and release DRB(s) in source cell

    f_EUTRA_SS_SRBs_DRBs_Reset(p_SourceCellId, -, -, -, v_CnfFlag); // @sic R5s100801 R5s150338: no reconfiguration of DRBs; R5s150922 - Additional changes: f_EUTRA_SS_SRBs_DRBs_Reset sic@

    // Step 14: Release DRX and MeasGap configuration

    f_EUTRA_SS_DrxCtrlConfig(p_SourceCellId, cs_DrxCtrl_None, -, v_CnfFlag);  // @sic R5s100801 sic@

    f_EUTRA_SS_MeasGapCtrlConfig(p_SourceCellId, cs_MeasGapCtrl_None, -, v_CnfFlag);  // @sic R5s110743 sic@
  }
 After change

      function f_EUTRA_508RRC_IntraLTE_HO_InterCell_Common(EUTRA_CellId_Type                         p_SourceCellId,

                                                       EUTRA_CellId_Type                         p_TargetCellId,

                                                       template (value) DL_DCCH_Message          p_RRC_ConnRecongMsg,

                                                       template (value) RachProcedureConfig_Type p_RachProcedureConfig,

                                                       template (value) RadioBearerList_Type     p_DrbConfigList := cs_DrbConfigList_DRB1,

                                                       C_RNTI                                    p_C_RNTI        := tsc_C_RNTI_Def2,

                                                       template (omit)  DciUlInfo_Type           p_DciUlInfo_TargetCell      := cs_DciInfo_CcchDcchDtchUL_Def,

                                                       template (value) UL_GrantConfig_Type      p_UL_GrantConfig_SourceCell := cs_UL_GrantConfig_Def,

                                                       template (value) UL_GrantConfig_Type      p_UL_GrantConfig_TargetCell := cs_UL_GrantConfig_Def,

                                                       template (present) UL_DCCH_Message        p_RRCConnReconfigComplete := cr_508_RRCConnectionReconfigurationComplete,

                                                       boolean                                   p_WaitForTAU := false) 

    runs on EUTRA_PTC

  { /* @sic R5s120709 function introduced to cope with "Full Handover" as needed for TC_13_4_1_5 sic@ */

    /* @sic R5s140147: additional parameter p_RRCConnReconfigComplete sic@ */

    var DRB_IdentityList_AllAndAMonly_Type v_DRB_IdentityList;

    var EUTRA_SecurityParams_Type v_Auth_Params := f_EUTRA_Security_Get();

    var NextHopChainingCount v_NextHopChainingCount := 0; //According to 36.508 cl. 4.6.1 RRCConnectionReestablishment

    var template (omit) MeasConfig v_MeasConfig := p_RRC_ConnRecongMsg.message_.c1.rrcConnectionReconfiguration.criticalExtensions.c1.rrcConnectionReconfiguration_r8.measConfig;

    var boolean v_MeasGapConfigIsPresent := isvalue(v_MeasConfig) and ispresent(v_MeasConfig.measGapConfig);

    var DrxCtrl_Type v_DrxCtrl;

    var boolean v_CnfFlag := tsc_NoCnfReq;  /* @sic R5s150922 - Additional changes: v_CnfFlag instead of v_CnfFlag sic@ */

    // Build DRB identity list from inputed radiobearer config list

    v_DRB_IdentityList := f_EUTRA_DRB_IdentityList(p_DrbConfigList);

    // Step 0: Configuration of DRBs at the target cell

    f_EUTRA_SS_CommonRadioBearerConfig(p_TargetCellId, p_DrbConfigList);       /* @sic R5s150338 - dynamic DRB configuration sic@ */

    // Step 1: Transfer of the PDCP Count for AM DRBs from source to target cell   @sic R5-120721 change 7 sic@

    f_EUTRA_SS_PdcpCount_Handover(p_SourceCellId, p_TargetCellId, v_DRB_IdentityList.AMonly);

    // Step 2: Tell the SS about the HO    @sic R5w100212 sic@

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetCellId, cas_PdcpHandoverInit_REQ(p_SourceCellId, p_TargetCellId)); // @sic R5s100556, R5s100560 sic@

    // Step 3: Configure C-RNTI based contention resolution in target cell

    f_EUTRA_SS_ConfigRachProcedure(p_TargetCellId, p_C_RNTI, p_RachProcedureConfig);

    // Save new C-RNTI in cell configuration for p_TargetCellId

    f_EUTRA_CellInfo_SetC_RNTI(p_TargetCellId, p_C_RNTI);

    // Step 4: activate RRC security at SS target cell for SRB 1, 2 and 1 AM DRB

    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest(p_TargetCellId,

                                                             v_Auth_Params,

                                                             v_NextHopChainingCount,

                                                             -,

                                                             v_DRB_IdentityList.AMonly);

    f_EUTRA_Security_Set(v_Auth_Params);

    // Step 5: configure DRX and MeasGap at the target cell (if necessary)

    if (f_EUTRA_CellInfo_DrxIsConfigured(p_SourceCellId)) {

      v_DrxCtrl := f_EUTRA_CellInfo_GetDrxCtrl(p_SourceCellId);

      f_EUTRA_SS_DrxCtrlConfig(p_TargetCellId, v_DrxCtrl);

    }

    if (v_MeasGapConfigIsPresent) {

      f_EUTRA_SS_MeasGapCtrlConfig(p_TargetCellId, cs_MeasGapCtrl_Config(valueof(v_MeasConfig.measGapConfig)));   // valueof to avoid compiler warnings: ok as v_MeasConfig.measGapConfig is checked to be present

    }

    // Step 6: stop periodic TA on source cell   @sic R5s100520 change 10 sic@

    f_EUTRA_SS_CommonCellConfig(p_SourceCellId, cas_ULGrantAllocation_REQ(p_SourceCellId, cs_TimingInfo_Now, -, cs_PUCCH_Synch_None, p_UL_GrantConfig_SourceCell));

    // Step 7: configure UL grant configuration on target cell    @sic R5s100515 sic@

    f_EUTRA_SS_CommonCellConfig(p_TargetCellId, cas_ULGrantAllocation_REQ(p_TargetCellId, cs_TimingInfo_Now, p_DciUlInfo_TargetCell, cs_PUCCH_Synch_None, p_UL_GrantConfig_TargetCell));

    f_IP_ChangeEutraCell(IP, p_TargetCellId);      /* @sic R5s150031 Change 1 sic@ */

    // Step 8: Transmit RRCConnectionReconfiguration/HO command to UE

    SRB.send(cas_SRB1_RrcPdu_REQ(p_SourceCellId, cs_TimingInfo_Now, p_RRC_ConnRecongMsg));

    // Step 9:

    SRB.receive(car_SRB1_RrcPdu_IND(p_TargetCellId, p_RRCConnReconfigComplete));       // @sic R5s100430  Change 2, R5s140147 sic@

    // Step 10: configure periodic TA on target cell    @sic R5s100525 change 1.3 sic@

    f_EUTRA_SS_CommonCellConfig(p_TargetCellId, cas_PUCCH_Synch_Config_REQ(p_TargetCellId, cs_TimingInfo_Now, cs_PUCCH_Synch_Auto_Def, tsc_NoCnfReq)); // @sic R5s100801 sic@

    // Step 11: Tell the SS about the HO    @sic R5w100212 sic@

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetCellId, cas_PdcpHandoverComplete_REQ(p_TargetCellId)); // @sic R5s100556, R5s100560 sic@

    // Step 12: configure RACH procedure for target cell back to default    @sic R5s100520 change 11 sic@

    f_EUTRA_SS_ConfigRachProcedure_Def(p_TargetCellId, -, -, -, -, -, v_CnfFlag);     /* @sic acc. to R5s100029 change 3 sic@ */

    if (not p_WaitForTAU) {              f_EUTRA_508RRC_IntraLTE_HO_InterCell_Common_step13to14(p_SourceCellId, p_TargetCellId);

      }
  }

function f_EUTRA_508RRC_IntraLTE_HO_InterCell_Common_step13to14(EUTRA_CellId_Type p_SourceCellId,       

                                                                EUTRA_CellId_Type p_TargetCellId) runs on EUTRA_PTC

  { 

    var boolean v_CnfFlag := tsc_NoCnfReq;  /* @sic R5s150922 - Additional changes: v_CnfFlag instead of v_CnfFlag sic@ */

    // Step 13: Re-establish SRBs and release DRB(s) in source cell

    f_EUTRA_SS_SRBs_DRBs_Reset(p_SourceCellId, -, -, -, v_CnfFlag); // @sic R5s100801 R5s150338: no reconfiguration of DRBs; R5s150922 - Additional changes: f_EUTRA_SS_SRBs_DRBs_Reset sic@

    // Step 14: Release DRX and MeasGap configuration

    f_EUTRA_SS_DrxCtrlConfig(p_SourceCellId, cs_DrxCtrl_None, -, v_CnfFlag);  // @sic R5s100801 sic@

    f_EUTRA_SS_MeasGapCtrlConfig(p_SourceCellId, cs_MeasGapCtrl_None, -, v_CnfFlag);  // @sic R5s110743 sic@  

  }
